Sunroof: A Monadic DSL to Generate JavaScript

Jan Bracker\textsuperscript{1,2} and Andy Gill\textsuperscript{1}

\textsuperscript{1} ITTC / EECS
The University of Kansas, Lawrence, KS 66045

\textsuperscript{2} Institut für Informatik
Christian-Albrechts-Universität, Kiel, Germany

Abstract. Sunroof is a Haskell-hosted Domain Specific Language (DSL) for generating JavaScript. Sunroof is built on top of the JavaScript monad, which, like the Haskell \textit{IO}-monad, allows access to external resources, but specifically JavaScript resources. As such, Sunroof is primarily a feature-rich foreign-function API to the browser’s JavaScript engine, and all the browser-specific functionality, including HTML-based rendering, event handling, and drawing to the HTML5 canvas.

In this paper, we give the design and implementation of Sunroof. Using monadic reification, we generate JavaScript from a deep embedding of the JavaScript monad. The Sunroof DSL has the feel of native Haskell, with a simple Haskell-based type schema to guide the Sunroof programmer. Furthermore, because we are generating code, we can offer Haskell-style concurrency patterns, such as MVars and Channels. In combination with a web-services package, the Sunroof compiler offers a robust platform to build interactive web applications.

Keywords: DSLs, JavaScript, Web Technologies, Cloud Computing

1 Introduction

JavaScript is an imperative language with access to a wide range of established and useful services like graphical canvases and event handling. It also provides features that are associated with functional languages, such as first-class functions. We want to express JavaScript in Haskell, adding use of Haskell’s static typing, and gaining access to JavaScript services in the browser from Haskell. Sunroof has the look and feel of monadic [22] Haskell code. As an example, consider this Haskell program, and the generated JavaScript:

```haskell
-- Haskell / Sunroof // JavaScript
jsCode :: JS t ()
jsCode = do
  name <- prompt "Your name?"
  var v0 = prompt("Your name?");
  alert("Your name: " <> name)
  alert("Your name: " + v0);
```

As can be seen, Sunroof code is monadic, and the JavaScript is a transliteration.
Sunroof’s approach has several advantages. First, Sunroof introduces a threading model and abstraction similar to Haskell’s. Second, we also utilize the static type system to support us when writing code. For example, the functions in our example above have the following signatures:

\[
\text{prompt} :: \text{JSString} \rightarrow \text{JS} \; \text{t} \; \text{JSString} \\
\text{alert} :: \text{JSString} \rightarrow \text{JS} \; \text{t} \; ()
\]

This allows the type checker to detect a wide class of malformed JavaScript programs. Third, using a deep embedding gives opportunity for optimizations when producing the actual JavaScript. At the same time, Sunroof offers an interface that is close to actual JavaScript making it easy to use. Finally, the interface is easily extendable through a foreign-function interface.

Fig. 1 shows Sunroof’s structure. We will cover each of the layers throughout the paper:

– Section 2 will discuss how we annotate JavaScript objects with types using wrappers and offer the possibility to add custom types later on.
– The semantics and implementation of the JS-monad is explained in Section 3.
– The special role of functions, continuations and how we model them as first-class values in Haskell and JavaScript will be covered in Section 4.
– The two threading models offered by Sunroof are explained in Section 5.
– Section 6 introduces Sunroof’s foreign-function interface.
– Translation of Sunroof to JavaScript is handled in Section 7. We will explain the compilation of selected language constructs. This is especially interesting in the light of our use of continuations and their translation to JavaScript.
– The ability to interleave Haskell and JavaScript computations as needed through the Sunroof server will be highlighted in Section 8.
– Section 9 will cover a small application written in Sunroof to survey how usable Sunroof is in the context of application development.

2 JavaScript Object Model

One goal of Sunroof is to use Haskell’s type system to increase the confidence of the expressed JavaScript. At the same time we cannot characterize all different types of objects in JavaScript, since users can create their own objects. Thus our system to type JavaScript needs to be extensible.
Our solution is to provide a basic Expression language to construct JavaScript expressions that have no associated type information. Simplified slightly, we have:

```haskell
data Expr
  = Lit String  -- Precompiled (atomic) JavaScript literal
  | Var Id      -- Variable
  | Apply Expr [Expr] -- Function application
...
```

In reality, we abstract `Expr` over the recursive type, to facilitate the usage of observable sharing [15] and allow sub-expression computations to be shared. This core expression type is then wrapped to represent a more specific type. Each of these wrappers implements the `Sunroof` type class.

```haskell
class SunroofArgument a => Sunroof a where
  box :: Expr -> a
  unbox :: a -> Expr
...
```

It marks these types as possible values in JavaScript. The `SunroofArgument` prerequisite permits them to be function arguments (Section 4).

An example of this is `JSString`, the representation of JavaScript strings.

```haskell
data JSString = JSString Expr
instance Sunroof JSString where
  box = JSString
  unbox (JSString e) = e
```

But what do we gain through a wrapper? We can provide specific functionality for each distinct type. Our example type `JSString` has a `Monoid` and an `IsString` instance that are not provided for other wrappers, e.g. `JSBool` or `JSNumber`. This is the same approach used by Svenningsson [29].

Table 1 gives a summary of the prominent Sunroof types. Some types involve phantom types to give more type safety [7]. The smooth embedding of booleans and numbers is done through the Boolean package [12].

Table 1 shows that most basic Haskell types have counterparts in Sunroof. To convert Haskell values into their counterparts, we provide the `SunroofValue` class.

```haskell
class SunroofValue a where
  type ValueOf a :: *
  js :: (Sunroof (ValueOf a)) => a -> ValueOf a
```

The type function `ValueOf` [6] provides the corresponding Sunroof type. `js` converts a value from Haskell to Sunroof. By design `SunroofValue` only provides instances for values that can be converted in a pure manner. Some types in JavaScript are referentially transparent according to `==`, while others, like general objects, are not. As an example, if you call `new Object()`, twice you get two
equivalent empty objects, but when compared by == they are different. They are not identical, because in this case reference equality is checked instead of value equality. We call this observable allocation and handle it as a side-effect which may only occur in the JS-monad.

3 The JavaScript Monad

The imperative nature and side-effects of JavaScript are modeled through the JS-monad. It resembles the IO-monad, but has an extra phantom argument [18] to decide which threading model is used. (For now we can ignore this extra phantom argument; it will be discussed in Section 5.)

The basic idea is that a binding in the JS-monad becomes an assignment to a fresh variable in JavaScript. This allows the results of previous computations to be passed on to later ones. Consider again the example from our introduction. The binding name is translated to the freshly generated variable v0. But this simple example displays a challenging problem. Where does v0 come from? Even though prompt can return a deep embedding expression, we want to reify the monad itself.

The bind inside the monadic do is fully polymorphic over a.

\[(\gg\gg) :: JS t a -> (a -> JS t b) -> JS t b\]

Thus we do not know how to create values of type a, that can be reified. To reify the monad itself, we want a constraint on a:

\[(\gg\gg) :: (Sunroof a) => JS t a -> (a -> JS t b) -> JS t b\]

where Sunroof constrains the bind to arguments for which we can generate a variable. Fortunately, it turns out that a specific form of normalization allows the type a to be constrained and JS to be an instance of the standard monad.
data JSI :: T -> * -> * where
  JS_Invoke :: (SunroofArgument a, Sunroof r) => a -> JSFunction a r -> JSI t r
  JS_Function :: (SunroofArgument a, Sunroof b) => (a -> JS A b) -> JSI t (JSFunction a b)
  JS_Branch :: (SunroofThread t, SunroofArgument a, Sunroof bool) => bool -> JS t a -> JS t a -> JSI t a
  JS_Assign_ :: (Sunroof a) => Id -> a -> JSI t ()

Fig. 2: Parts of the JavaScript instruction data type (JSI).

class [26]. To summarize the technique, a deep embedding of a monad is normalized using the monad laws, and as a consequence, constraints, like the ability to reify, can be internalized. Through this keyhole of monadic reification, the entire Sunroof language is realized.

The Sunroof normalization is done using Operational [3, 2]. It provides the Program-monad that can be equipped with custom primitives. We represent these primitives with the JavaScript instruction (JSI) type shown in Fig. 2. It represents the abstract instructions that are sequenced inside the Program-monad. The parameter \( t \) in \( JSI t a \) represents the threading model. The type \( a \) represents the primitive’s return value. \( JSInvoke \) calls a function that has been created with \( JSFunction \). Branches are represented with \( JSBranch \). Assignments to a variable are represented by \( JSAssign_ \).

Without the alternative threading model, the normalization through Operational would be enough to offer a monad suitable for Sunroof. Due to our threading plans, there is more than just normalization going on behind the scenes. The JS-monad is a continuation monad over the Program-monad.

data JS :: T -> * -> * where
  JS :: ((a -> Program (JSI t) ()) -> Program (JSI t) ()) -> JS t a

Thus, the internal monad instance used is the standard implementation of a continuation monad.

4 Functions and Continuations

Functions are first class values in Haskell and JavaScript. Sunroof represents function objects with the type \( JSFunction \alpha \beta \), which resembles a function \( \alpha \to \beta \) in JavaScript. Rather than directly support partial application, and transliterate curried Haskell into uncurried JavaScript, we only permit uncurried functions. Like JavaScript, currying can be encoded if needed. We introduced \( SunroofArgument \), a superclass of \( Sunroof \), to constrain the types that may be used as arguments for functions.
This class supports converting each argument into its expression through `jsArgs` to supply the arguments to a function call. `jsValue` generates unique names for each argument, which is needed when compiling the function itself to a value.

Now we can provide more than one argument by providing `SunroofArgument` instances for tuples of Sunroof values.

```haskell
instance (Sunroof a, Sunroof b) => SunroofArgument (a,b) where
    jsArgs (a,b) = [unbox a, unbox b]
    jsValue = liftM2 (,) jsVar jsVar
```

Remember that each `Sunroof` value already has to be a `SunroofArgument`, which enables us to pass a single argument to a function.

We can create function objects with the `function` combinator.

```haskell
function :: (SunroofArgument a, Sunroof b) => (a -> JS A b) -> JS t (JSFunction a b)
```

As a function can have side-effects, its computation and result have to be expressed in the JS-monad. The creation of a function is considered a side-effect, due to observable allocation.

Function application is done through the `apply` or `$$` combinator, they are synonyms. Functions can only be applied in the JS-monad, since they can have side-effects.

```haskell
apply, ($$) :: (SunroofArgument a, Sunroof b) => JSFunction args ret -> args -> JS t ret
```

Creation and application are implemented using the `JSFunction` and `JS_Invoke` instructions introduced in Fig. 2.

`JSContinuation α` is used to model continuations in the JS-monad. It was introduced to work with continuations in Sunroof. Technically they are only specializations of functions, but restricted to a specific threading model. Continuations are meant to be a representation of side-effects – ongoing computations inside the JS-monad – and may not terminate, so they do not return a value. As with functions, there is a combinator to create and apply a continuation.

```haskell
continuation :: (SunroofArgument a) => (a -> JS B ()) -> JS t (JSContinuation a)
goto :: (SunroofArgument a) => JSContinuation a -> a -> JS t b
```

The presented `goto` should not be considered harmful [10]. It calls a continuation, as `apply` calls functions. The difference is that a call to `goto` will never return,
as it executes the given continuation and abandons the current one. This means \texttt{goto} is fully polymorphic on its (unused and unusable) result.

Access to the current continuations is given through the powerful call-with-current-continuation combinator \texttt{callcc}.

\begin{verbatim}
callcc :: SunroofArgument a
    => (JSContinuation a -> JS B a) -> JS B a
\end{verbatim}

The current continuation models everything that would usually happen after the call to \texttt{callcc}.

\begin{verbatim}
callcc f = JS $ \ k \rightarrow unJS
    (continuation (\a \rightarrow JS $ \ _ \rightarrow k a) >>= f) k
\end{verbatim}

The implementation of \texttt{callcc} is interesting, because it shows how the \texttt{Program}-continuation is translated into a \texttt{JSContinuation} that is passed to the given function \texttt{f}. Section 5 will show why this function is important for Sunroof and what it is used for.

Functions and continuations are similar and connected to each other, as can be seen in Fig. 3. We can go back and forth between the Haskell and the Sunroof representation of a function or continuation. But once a function is specialized to a continuation, it is not possible to go back, because continuations only model the side-effect, but do not return anything.

5 Threading Models

JavaScript uses a callback centric model of computation. There is no concurrency, only a central loop that executes callbacks when events occur.

In contrast, Haskell has real concurrency and wide-spread abstractions for synchronization, e.g. \texttt{MVars} and \texttt{Chans} [17]. So the question arises: do we generate atomic JavaScript code, and keep the callback centric model, or generate JavaScript using CPS [8], and allow for blocking primitives and cooperative concurrency. The latter, though more powerful, precluded using the compiler to generate code that can be cleanly called from native JavaScript. Both choices had poor consequences.
We decided to explicitly support both, and make both first-class threading strategies in Sunroof. In terms of user-interface, we parameterize the JS-monad with a phantom type that represents the threading model used, with A for Atomic, and B for Blocking threads. Atomic threads are classical JavaScript computations that cannot be interrupted and actively use the callback mechanism. Blocking threads can support suspending operations and cooperative concurrency abstractions as known from Haskell. By using phantom types, we can express the necessary restrictions on specific combinators, as well as provide combinators to allow both types of threads to cooperate.

The blocking model hides the callback mechanism behind abstractions. This implies that every atomic computation can be converted into a blocking computation. liftJS achieves this.

\[
liftJS :: \text{Sunroof } a \Rightarrow \text{JS } A \ a \rightarrow \text{JS } t \ a
\]

When suspending, we register our current continuation as a callback to resume later. This gives other threads (registered continuations) a chance to run. Of course, this model depends on cooperation between the threads, because a not terminating or suspending thread will keep others from running.

There are three main primitives for the blocking model:

\[
\begin{align*}
\text{forkJS} & : \text{SunroofThread } t1 \Rightarrow \text{JS } t1 \ () \rightarrow \text{JS } t2 \ () \\
\text{threadDelay} & : \text{JSNumber} \rightarrow \text{JS } B \ () \\
\text{yield} & : \text{JS } B \ ()
\end{align*}
\]

They can all be seen as analogues of their IO counterparts. forkJS resembles forkIO. It registers the given computation as a callback. yield suspends the current thread by registering the current continuation as a callback, giving other threads time to run. threadDelay is a form of yield that sets the callback to be called after a certain amount of time. We rely on the JavaScript function window.setTimeout \[1\] to register our callbacks.

The class SunroofThread offers functions to retrieve the current threading model (evalStyle) and to create a possible blocking computation (blockableJS).

\[
\begin{align*}
\text{class SunroofThread } (t :: T) \text{ where} \\
\text{evalStyle} & : \text{ThreadProxy } t \rightarrow T \\
\text{blockableJS} & : (\text{Sunroof } a) \Rightarrow \text{JS } t \ a \rightarrow \text{JS } B \ a
\end{align*}
\]

Based on these primitive combinators, we also offer a Sunroof version of MVar and Chan: JSMVar and JSChan.

\[
\begin{align*}
\text{newMVar} & : (\text{SunroofArgument } a) \Rightarrow a \rightarrow \text{JS } t \ (\text{JSMVar } a) \\
\text{newEmptyMVar} & : (\text{SunroofArgument } a) \Rightarrow \text{JS } t \ (\text{JSMVar } a) \\
\text{putMVar} & : (\text{SunroofArgument } a) \Rightarrow a \rightarrow \text{JSMVar } a \rightarrow \text{JS } B \ () \\
\text{takeMVar} & : (\text{SunroofArgument } a) \Rightarrow \text{JSMVar } a \rightarrow \text{JS } B \ a \\
\text{newChan} & : (\text{SunroofArgument } a) \Rightarrow \text{JS } t \ (\text{JSChan } a) \\
\text{writeChan} & : (\text{SunroofArgument } a) \Rightarrow a \rightarrow \text{JSChan } a \rightarrow \text{JS } t \ () \\
\text{readChan} & : (\text{SunroofArgument } a) \Rightarrow \text{JSChan } a \rightarrow \text{JS } B \ a
\end{align*}
\]
Note that the types reflect if a specific operation can block. For example, `writeChan` can never block, so you can use either threading model, but `readChan` may block, so uses the B threading model.

6 Foreign Function Interface

Sunroof also offers a foreign function interface, which enables us to easily access predefined JavaScript. There are four core functions:

```haskell
fun :: (SunroofArgument a, Sunroof r) => String -> JSFunction a r
object :: String -> JSObject
new :: (SunroofArgument a)
    => String -> a -> JS t JSObject
invoke :: (SunroofArgument a, Sunroof o, Sunroof r)
    => String -> a -> o -> JS t r
```

`fun` is used to create Sunroof functions from their names in JavaScript. This can happen in two ways: either to call a function inline, or to create a real binding for that function. As an example, the `alert` function can be called in line through `fun "alert" $$ "text"`, or you can provide a binding in form of a Haskell function for it.

```haskell
alert :: JSString -> JS t ()
alert s = fun "alert" $$ s
```

Objects can be bound through the `object` function, e.g. the `document` object is bound through `object "document"`. Constructors can be called using `new`. To create a new object you would call `new "Object" ()`.

We can call methods of objects through `invoke`. Again, this can be used inline and to create a real binding. An inline use of this to produce `document.getElementById("id")` would look like this:

```haskell
object "document" # invoke "getElementById" "id"
```

where `#` is just a flipped function application. To provide a binding to the `getElementById` method, one can write:

```haskell
getElementById :: JSString -> JSObject -> JS t JSObject
getElementById s = invoke "getElementById" s
```

Providing actual bindings ensures that everything is typed correctly and prevents the need to resolve ambiguities through large type annotations inside of code.

The current release of Sunroof already provides bindings for most of the core browser API, the HTML5 canvas element, and some of the JQuery API.
7 The Sunroof Compiler

How do we compile Sunroof? Through the JS-monad we produce a `Program (JSI t) ()` instance. We translate such a program into a list of statements `_stmt` by matching over the JSI constructors.

```haskell
data Stmt = AssignStmt Rhs Expr -- Assignment
  | ExprStmt Expr -- Expression as statement
  | ReturnStmt Expr -- Return statement
  | IfStmt Expr [Stmt] [Stmt] -- If-Then-Else statement
...
```

The constructors of `Stmt` directly represent the different statements you can write in JavaScript. Operationals [3, 2] `Program` type is abstract and has to be converted to a `ProgramView` instance to work with. It provides the `:>>=` and `Return` constructors to pattern match on and translate the instructions inside.

```haskell
compile :: Program (JSI t) () -> CompM [Stmt]
compile p = eval (view p)
 where eval :: ProgramView (JSI t) () -> CompM [Stmt]
...
```

The `CompM` state monad provides the compiler options and fresh variables.

To get started, `JS_Assign_` is translated in the following manner. Recall from Fig. 2 that it takes a variable name and the value to assign as arguments.

```haskell
eval (JS_Assign_ v a :: k) = do
  (stmts0, val) <- compileExpr (unbox a) -- Expr -> CompM ([Stmt], Expr)
  stmts1 <- compile (k ()
  return ( stmts0 ++ [AssignStmt (VarRhs v) val] ++ stmts1)
```

First, we compile the expression `a` to assign into a series of statements that compute it together with the result value `val`. Next, we compile what happens after the assignment. Since the assignment produces unit, we can pass that to `k`. In the end, we concatenate all statements with the assignment in between.

Now we will look into the translation of branches. Recall the JSI constructor for branches from Fig. 2:

```haskell
JS_Branch :: (SunroofThread t, SunroofArgument a, Sunroof bool) => bool -> JS t a -> JS t a
...
```

Fig. 4 shows how to compile a `JS_Branch`. We generate the statements for the branching condition. Then we generate unique variables for each of the returned values in either branch and use `bindResults` to assign them to the new variables. The function `extractProgramJS` passes the result of a computation `m` into the function `f` and closes the continuation in JS with `return`. The result is a `Program` containing all instructions of `m >>= k`.

```haskell
extractProgramJS :: (a -> JS t ()) -> JS t a -> Program (JSI t) ()
extractProgramJS k m = unJS (m >>= k) return
```
eval (JS_Branch b c1 c2 :>>= k) = do
  (src0, res0) <- compileExpr (unbox b)
  res :: a <- jsValue
  let bindResults :: a -> JS t ()
      bindResults res' = sequence_ [ single $ JS_Assign_ v (box $ e :: JSObject)
                                  | (Var v, e) <- jsArgs res 'zip' jsArgs res' ]
  src1 <- compile $ extractProgramJS bindResults c1
  src2 <- compile $ extractProgramJS bindResults c2
  rest <- compile (k res)
  return (src0 ++ [ IfStmt res0 src1 src2 ] ++ rest)

Fig. 4: Naive translation of branches in Sunroof.

After the compilation of both branches, we translate the rest and create the list of statements in a canonical fashion.

This compilation strategy works perfectly if we are in the atomic threading model, but can fail when inside the blocking model. In the blocking model either branch may involve continuations. Each continuation describes the rest of the computation up to the end of that branch. It also captures the assignments at the end of the branch. Assignments captured inside a continuation are not visible outside of it. That means the variables are not visible after the branch and the code inside of rest refers to them although they are not defined in that scope.

Therefore, within the blocking threading model, we have to handle branches differently:

eval (JS_Branch b c1 c2 :>>= k) =
  case evalStyle (ThreadProxy :: ThreadProxy t) of
    A -> compileBranch_A b c1 c2 k
    B -> compileBranch_B b c1 c2 k

The call to compileBranch_A executes our naive definition from Fig. 4.

compiledBranch_B b c1 c2 k = do
  fn_e <- compileContinuation $ \a -> blockableJS $ JS $ \k2 -> k a >>= k2
  fn <- newVar
  (src0, res0) <- compileExpr (unbox b)
  src1 <- compile $ extractProgramJS (apply (var fn)) c1
  src2 <- compile $ extractProgramJS (apply (var fn)) c2
  return ( [mkVarStmt fn fn_e] ++ src0 ++ [ IfStmt res0 src1 src2 ] )

We can see that the rest of our computation is captured in the continuation \( fn_e \). It takes the results of a branch as arguments. A new variable \( fn \) is used to share the continuation in both branches. The key difference is the parameter to extractProgramJS. Instead of creating bindings to variables, we apply the produced continuation to the returned values. That passes them to the ongoing
computation. Of course, we could compile every branch with the continuation variant, but this would unnecessarily obfuscate the produced code.

The compiler offers two functions to compile either threading model:

```haskell
sunroofCompileJSA :: Sunroof a
  => CompilerOpts -> String -> JS A a -> IO String
sunroofCompileJSB :: CompilerOpts -> String -> JS B () -> IO String
```

Notice that for the blocking threading model only unit may be returned due to their continuation-based nature. The string argument provides the name of the variable to which to bind the computation result.

## 8 The Sunroof Server

The Sunroof server provides infrastructure to send arbitrary pieces of JavaScript to a calling website for execution. So, it is possible to interleave Haskell and JavaScript computations with each other as needed. The three major functions provided are `sunroofServer`, `syncJS` and `asyncJS`.

```haskell
type SunroofApp = SunroofEngine -> IO ()
sunroofServer :: SunroofServerOptions -> SunroofApp -> IO ()
syncJS :: SunroofResult a
  => SunroofEngine -> JS t a -> IO (ResultOf a)
asyncJS :: SunroofEngine -> JS t () -> IO ()
```

`sunroofServer` starts a server that will call the given callback function for each request. `syncJS` and `asyncJS` allow the server to run Sunroof code inside the requesting website. `asyncJS` executes it asynchronously without waiting for a return value. In contrast to that, `syncJS` waits until the execution is complete and then sends the result back to the server. It is converted into a Haskell value that can be processed further. Values that can be converted to a Haskell type after a synchronous call implement the `SunroofResult` class. It maps the Sunroof type to a corresponding Haskell type through a type function and provides a function to convert the data to that type.

## 9 Case Study - A Small Calculator

To see how Sunroof works in practice, we will look into the experience we gathered when writing a small calculator for arithmetic expressions (Fig. 5). We use Sunroof to display our interface and the results of our computation. Haskell will be used to parse the arithmetic expressions and calculate the result. The Sunroof server will be used to implement this JavaScript/Haskell hybrid.

The classical approach to develop an application like this would have been to write a server that provides a RESTful interface and replies through a JSON
How does Sunroof improve or change this classical structure? First of all, in Sunroof you write the client-side code together with your server application within Haskell. In our example, all code for the server and client is in Haskell. The control logic for the client side is provided through the server. This leads to a tight coupling between both sides. This also shows how Sunroof blurs the border between the server and client side. You are not restricted by an interface or language barrier. If you need the client to do something, you can just send arbitrary Sunroof code to execute in the client.

Table 2 contains a few statistics about the size of the code in each part of the client.

The client-server response loop shuffles new input to the server and executes the response in the client. Data conversion is needed, because pure Haskell data types cannot be handled in Sunroof and vice versa. There still exists a language barrier between JavaScript and Haskell. Code to convert between two essentially equal data structures on each side must be written, as well as representations of

---

![Diagram](image-url)

**Fig. 5:** The example application running on the Sunroof server.

**Fig. 6:** Classical structure and Sunroof structure of a web application.

data structure. The client side of that application would have been written in JavaScript directly. This can be seen in Fig. 6.
Haskell structures in Sunroof. However, there is great potential in automatically generating this code using techniques such as template Haskell [27].

The code for displaying the results is basically a transliteration of the JavaScript that you would write for this purpose. The transliteration used here is not very appealing. In the future, this code can be generated through higher-level libraries. Sunroof is intended to deliver a foundation for this purpose.

The rest of our code to parse the arithmetic expression and calculate results is classical Haskell code.

### Table 2: Lines of code needed for the example.

<table>
<thead>
<tr>
<th>Part of Application</th>
<th>Lines of Code</th>
<th>Percentage</th>
</tr>
</thead>
<tbody>
<tr>
<td>Response loop</td>
<td>25</td>
<td>6.5%</td>
</tr>
<tr>
<td>Data conversion</td>
<td>85</td>
<td>22.0%</td>
</tr>
<tr>
<td>Rendering</td>
<td>190</td>
<td>49.5%</td>
</tr>
<tr>
<td>Parsing and interpretation</td>
<td>85</td>
<td>22.0%</td>
</tr>
</tbody>
</table>

10 Related Work

There have been several attempts to translate Haskell to JavaScript. Prominent ones are the compiler backends for UHC [28] and GHCJS [24]. There are also projects like Fay [11] that compile subsets of Haskell to JavaScript or JMacro [5] which use quasiquote [21] to embed a custom-tailored language into Haskell code. At the same time there are also projects like CoffeeScript [4] or LiveScript [23] to build custom languages that are very similar to JavaScript but add convenient syntax and support for missing features.

Our approach to cooperative concurrency through continuations in JavaScript has has been used before [9, 25]. To our knowledge, creating a direct connection between Haskell and JavaScript continuations has not been attempted before.

Deep embeddings of monads based on data structures have been used before in Unimo [19] and Operational [3, 2]. The specific approach Sunroof takes by using GADTs has been discussed by Sculthorpe et al. [26] in detail.

The Sunroof server does not have the aim to provide a full-featured web framework, as HAppS, Snap or Yesod do. It only provides the infrastructure to communicate with the currently calling website through the Kansas comet [16] push mechanism [20]. Although all of the frameworks mentioned above would be able to implement this technique, to our knowledge, none of them has yet.

To our knowledge, Sunroof is the only library that supports generation of JavaScript inside of Haskell using pure Haskell in a type-safe manner. All other approaches discussed above either require a separate compilation step or introduce new syntax inside of Haskell. There also is an effort to generalize Active [30], a library for animations, and implement a backend based on Sunroof [14].
11 Conclusion

Sunroof took the key idea of monad reification and successfully created the JS-monad to describe computations in JavaScript. This reification work was started by Farmer and Gill [13], with the observation of the possibility to reify a JS monad. This paper documents the work since this initial prototype. By stronger usage of types, and the concepts of JSFUncti\on and JSContinuation, there now is a stronger connection between functions in the JavaScript and the Sunroof language space (Fig. 3). It is possible to go back and forth between both worlds. Combining both concepts, functions and the JS-monad, we were able to create a second implementation of the monad, this time based on the direct translation of continuations from Haskell to JavaScript. It enabled us to build a blocking threading model on top of JavaScript that resembles the model already known from Haskell. Based on this model and the provided abstraction over continuations, we can use primitives like forkJS or yield. Higher-level abstractions like JSMVar and JSChan are also available.

12 Acknowledgment

We want to thank Conal Elliott for his support in adapting the Boolean package [12] and helping us to extend it with support for deeply embedded numbers.

References